
About These Slides 

These slides are intended for self-study. They have not 

been presented in a time slot of sufficient duration. 



Acknowledgement 
  

This presentation benefitted from expertise 

of Cynthia Zender and Chevell Parker  

at SAS Institute. 
 

Search on their names and any non-DDE tools 

used here for more information. 
 

For ExcelXP, search for papers of Vince 

(or Vincent) Del Gobbo also at SAS Institute. 



Purpose of Presentation 
  

• Problem: 

• Too many tool choices 

• Users really want only ONE choice 

• Variety of documentation resources 

• (Partial) Solution Here: 

• ONE place to compare tools 

• to assist tool selection based on need 

• For a simple unformatted data dump, 

use PROC EXPORT or the LIBNAME 

engine, not to be discussed here 
 



Content of Presentation 
  

• Best Reason to use ODS TableEditor TagSet: 

  Pivot Tables 

• ODS HTML is adequate for simple needs 

• Significant Advantages of ODS tagsets 

  over the simpler solution of ODS HTML 

• Frozen Headers, Filters, AutoFit Rows/Cols 

• DDE (Dynamic Data Exchange) 

• Multiple WorkSheets per WorkBook 

• WorkSheet with Multiple Graphs  

• WorkSheet with Multiple Tables  

• WorkSheet with Multiple Tables and Graphs 



More . . . 
  

In the Conference Proceedings & Tools, find: 

• ODS Code Examples Used Here 

• Other ODS SAS-to-Excel Resources— 

Tips, References, Sample Program SetUp 

• Detailed DDE Code Example 

• Stepwise use of DDE, showing effect of 

each command on spreadsheet being built 

• An expansive DDE paper 

• DDE toolkit of 60 macros, 28 sample 

programs, and supporting files 



Related SAS Institute Training 

Exporting SAS Data Sets 

and Creating ODS Files for Microsoft Excel 
 

Exported data is unformatted 
 

DDE tool not included in the course 



Comment 

Graphs used here are for proof of concept, 

but are not meant as good graphic design. 



Ways to Do SAS to Excel 

with highly formatted output 
  

• DDE (Dynamic Data Exchange)* 

• ODS HTML 

• ExcelXP tagset 

• MSOffice2K tagset (less capability) 

• MSOffice2K_x tagset 

• TableEditor tagset** 

*SAS must be running on Windows 

**JavaScript must be enabled 



About MSOffice2K 
  

• Less capability than MSOffice2K_x, ExcelXP, 

TableEditor, and DDE 

• If your needs are nominally met by ODS HTML, 

then MSOffice2K will do a nicer job 

• However, MSOffice2K is not within the scope of 

this presentation & the project that led to it 

• If interested, please investigate via search at 

support.sas.com, www.lexjansen.com, or the 

whole wide world via the browser of your choice 

 



Why Use MSOffice2K? 
  

“For simple HTML-based output, use ODS 

MSOFFICE2K to create HTML tags that Microsoft 

uses and to get an XLS file that you can open 

with Excel and still respects the SAS ODS style 

you have chosen.” 

— My quotation of an expert, but paraphrased 



 Feature  TableEditor*  ExcelXP  MSOffice2K_x     HTML       DDE** 

 Pivot Table  Yes  No  No  No 

 requires an 

Excel macro 

 AutoFit 

 Columns  By Default*   No  By Default* By Default  Yes 

 Filters 

 Yes, but All 

 Columns only  Yes 

 Yes, but All 

 Columns only  No  Yes 

 Freeze Panes  Row 1 only  Yes  Yes  No  Yes 

 Side-By-Side 

 Elements in 

 WorkSheet  Yes  No  Yes  No  Yes 

 Multi-Sheet 

 WorkBook  Yes  Yes, Easy  Yes  No  Yes 

 Graphs  Yes  No  Yes  Yes  Yes 

 Titles 

 Not with 

 other features   Yes  Yes  Yes  Yes 

Comparison of Tool Alternatives 

*Headings of columns autofit by default can be overlaid by filter buttons 

*Javascript must be enabled **SAS must be running on Windows 



ODS TagSets 
  

• tagsets updated without new release of SAS 

• part of Base SAS 

• can be user installed 

• documentation  

• in many places 

• no manuals 

• get doc in SAS log by including 

  options(doc='help') in your ODS statement, 

  but the result might suggest more resources 

• for more information on the matters above, 

  see support.sas.com, or email request 

  my “Tips and Suggested Related Reading” 

 



        Why care about TagSets? 
  

If you don’t use DDE, there are 

things ODS HTML cannot do. 

(DDE can do more than TagSets 

but it might not be an option for you) 

 

 



TableEditor TagSet 

IS 

the only ODS solution 

to create Pivot Tables 

 

 



The Best Reason 

to use 

TableEditor TagSet 

 

 





ods results; /* MUST BE TURNED ON 

                       so you can get and respond to prompts */ 

ods tagsets.tableeditor 

file="%sysfunc(getoption(work))\temp.html" 

options( 

pivotrow="product_line" 

pivotcol="quarter" 

pivotdata="profit" 

pivotpage="year" 

auto_excel="yes" 

excel_save_file="C:\\SomeFolder\\PivotTbl.xls"); 

/* Note mandatory double back slash \\ in the path. */ 

proc print data=sashelp.orsales; run; 

ods tagsets.tableeditor close; 

/* MORE pivot table options are available */ 



ods results; /* MUST BE TURNED ON 

                       in your SAS code 

                       so you can get and respond to prompts */ 

/* ALSO, in SAS Display Manager, under Tools > 

Options > Preferences > Results 

on the Results tab, in the Results options section, 

you MUST Check “View results as they are generated” 

so you can get and respond to prompts. 

On the Results tab, my other preferences are: 

1. Uncheck “Create listing” and “Create HTML” and 

“Use ODS Graphics” (so I can control that through my 

code with no possible interference from SAS defaults) 

2. For “View results using:”, select “Internal Browser”. 

(Using an external browser might cause problems, 

depending on which one you use. Step 1 of the ODS 

TableEditor run opens an intermediate HTML file.)  



/* Two-Step Process: 

1. Output a web page 

2. Convert it to an Excel file */ 

 

ods tagsets.tableeditor 

file="%sysfunc(getoption(work))\temp.html" 

options( 

 < formatting options here > 

auto_excel="yes" 

excel_save_file="C:\\SomeFolder \\PivotTbl.xls"); 

/* Note mandatory double back slash \\ in the path. */ 



This two-step process cannot be run in batch. 

It requires responses (both “Yes”) to two prompts. 

Excel file has already been created when second prompt 

appears. This is the first prompt: 



This two-step process cannot be run in batch. 

Excel file has already been created when the second 

prompt appears. Below is an example of the second 

prompt, where most of it has been clipped off. The text 

can vary. It is computer-specific and path-specific. The 

“error” is always harmless, at least in my experience. 



TableEditor for Pivot Table 
  

• Benefit: 

• Meets a very popular desire 

• Coding is easy 

• Limitation: 

• Code must be run in Display Manager 

  so that you can respond to prompts 

• Cannot be run in batch hands-off 

• Cannot be run as a scheduled job 



Comparison of 

ODS TagSets, ODS HTML, and 

Dynamic Data Exchange (DDE) 

 

 

 

 

 



     

  Easiest Solution for Minimal Need 
 

- If need is limited to: 

  - font, size, color for data & headings 

  - title(s) (and footnotes) 

  - possibly a hyperlink in a title/footnote 

- use ODS HTML with filetype XLS 

- use any ODS style that you like 

- optionally override with STYLE(. . .)=[ . . . ] 

  in PROC PRINT statements 

- you might need to use some HTML tags 

 
 



For more on use of STYLE(. . .)=[ . . . ] 

to customize a complex table, 

see my WIILSU Proceedings-Only paper: 

Using Color to Communicate, 

Not to Decorate 
 

Also, to color-code spreadsheet data 

when NOT using DDE, see 

Color-Coding Your Data (NOT “Traffic-

Lighting”) 

in this same paper. 

 

 
 



ODS HTML 





ods noresults; /* prevents pop-up about Save or Open */ 

ods graphics / reset=all; ods graphics off; 

goptions reset=all; 

ods html style=Minimal path=“C:\Folder" file="File.xls"; 

title1 '<td COLSPAN=16><font color=blue size=4 face=Georgia> 

HTML font controls . . .</font></td>'; 

  /* HTML font sizes are 1,2,3,4,5,6,7, which are not point sizes */ 

 < other TITLE statements here > 

title4 "<td COLSPAN=16><a href='http://www.wiilsu.org'> 

<font size=4 face=Georgia color=red>Go to Wisconsin Illinois 

SAS Users Home Page</font></a></td>"; 

  /* title4 justify=left height=16pt color=red 

     link='http://www.wiilsu.org'  'Go to . . . '; 

     The above way to create a hyperlink in a title 

     is not applicable with ODS HTML. */ 

title5 . . . ; 

proc print data=sashelp.class noobs; run; 

ods html close; 



title1 '<td COLSPAN=16> 

<font color=blue size=4 face=Georgia> 

HTML font controls are used for this TITLE statement 

</font> 

</td>'; 
 

- HTML font sizes are 1,2,3,4,5,6,7, 

which are not point sizes. 
 

- Without HTML column spanning, 

worksheet titles always are in Column A, 

which is stretched. 
 

- With HTML column spanning, 

worksheet titles always start in Column B. 

 

 

 

 

 



  ODS HTML or ODS tagset output 

  with filetype XLS 
 

      When you open such an XLS file, 

  you get misleading intimidating message: 
 

      The file you are trying to open, 'FileName.xls', 

   is in a different format than specified by the file extension. 

   Verify that the file is not corrupted 

   and is from a trusted source before opening the file. 

   Do you want to open file now? 
 

       Yes No Help 

 

 

 

 
 



           

WHY you get the nuisance message 

 

Your file is not “true, binary” Excel output. 

The message is warning you that 

file contents do not match the file extension. 

 

A SAS Note explains this in more detail. 

(You can change the Windows registry 

to suppress the message.) 

See: http://support.sas.com/kb/31/956.html 
 

 

 



 

     

 With ODS ExcelXP, 

 if you use filetype XML instead of XLS 
 

    You will AVOID that message 

 

 
 

 

 

 
 

 

 



     

    Frequent-Use Features in Excel 
 

- Titles 

- AutoFit Column Widths & Row Heights* 

- Frozen Headers* 

- Frozen RowHeaders* 

- Filters* 

- Hyperlink in a title (and/or footnote) 

- Zoom* 

* Not available with ODS HTML 

* Only available with DDE 

 
 

 



 

          Effect of Each TagSet 
 

- It implements each feature IF supported, 

  but possibly somewhat differently 
 

- For the same ODS style, each tagset 

  renders appearance of that style differently 
 

- I have used STYLE=Minimal 

 

 

 
 

 

 



             TagSet Support for Filters 

ExcelXP – All Columns or Selective, Easy 

MSOffice2Kx – Only All Columns 

TableEditor – All Columns or Selective* 

*Selective-Columns Option Is Defective 

 

 

 



        Let’s Compare Results for 
  

    ExcelXP 

    MSOffice2K_x 

    TableEditor 

    Dynamic Data Exchange (DDE) 

 

 

 

 



ExcelXP: Freeze 8 Hdrs (top rows) & 3 RowHdrs,4 Filters 



MSOffice2K_x: Freeze 7 Hdrs (top rows) & 3 RowHdrs, Filter All 



TableEditor: Freeze Top Row, Filters on All Columns 

No Title Lines Possible with Filters and/or Frozen Top Row 

Columns are AutoFit, but filter buttons partially overlay headings 



DDE: AutoFit Cols, Freeze 5 Hdrs & 3 RowHdrs, 4 Filters 



         

  Now Let’s Compare Results 

  and See the Code for: 
  

  ExcelXP 

  MSOffice2K_x 

  TableEditor 

  StepWise Demo Run of Code for 

  DDE is at the end of these slides 

 



ExcelXP: Freeze 8 Hdrs (top rows) & 3 RowHdrs,4 Filters 



ods noresults; /* prevents pop-up about Save or Open */ 

ods graphics / reset=all; ods graphics off; 

goptions reset=all; 

ods tagsets.ExcelXP path="C:\SomeFolder" 

style=minimal 

  file="SomeFileName.xml"   

  /* can use file extension xls instead, 

  but it causes a misleading message at file open time */ 

  options( 

wraptext="no" /* to solve problem */ 

zoom="65“ embedded_titles="yes" 

frozen_headers=“8"  frozen_rowheaders="3" 

autofilter="1-4" 

sheet_name="ExcelXP Features Demo" ); 

 < PROC PRINT STEP HERE > 

ods tagsets.ExcelXP close; 



 /* PROC PRINT STEP */ 
 

proc print data=sashelp.shoes noobs; 

title1 justify=left height=14pt color=blue 'title1 text'; 

 < other TITLE statements here > 

title6 justify=left height=16pt color=blue 

link='http://www.wiilsu.org'  

'Go to Wisconsin Illinois SAS Users Home Page'; 

  /* title6 "<a href='http://www.wiilsu.org'>Go to 

Wisconsin Illinois SAS Users Home Page</a>"; 

  The above way to create a hyperlink in the title 

  does not work with this tagset. */ 

id region subsidiary product; 

run; 



MSOffice2K_x: Freeze 7 Hdrs (top rows) & 3 RowHdrs, Filter All 



ods noresults; /* prevents pop-up about Save or Open */ 

ods graphics / reset=all; ods graphics off;  

goptions reset=all; 

ods tagsets.msoffice2k_x path="C:\SomeFolder" 

  style=minimal 

  headtext="<style> h1 {margin:0}</style>"  

  /* above fix prevents blank lines between title lines */ 

  file="SomeFileName.xls" 

  options( 

zoom="120“ 

orientation="landscape" 

frozen_headers="7"  frozen_rowheaders="3" 

autofilter="yes" 

sheet_name="MSOffice2K_x Features Demo" ); 

 < PROC PRINT STEP HERE > 

ods tagsets.msoffice2k_x close; 



/* PROC PRINT STEP */ 
 

proc print data=sashelp.shoes noobs; 

title1 justify=left height=14pt color=blue 'title1 text'; 

 < other TITLE statements here > 

title5 justify=left "<a href='http://www.wiilsu.org'> 

Go to Wisconsin Illinois SAS Users Home Page</a>"; 

  /* title5 justify=left link='http://www.wiilsu.org'  

  'Go to Wisconsin Illinois SAS Users Home Page'; 

  The above way to create a hyperlink in the title 

  does not work with this tagset. */ 

id region subsidiary product; 

run; 



TableEditor: Freeze Top Row, Filters on All Columns 

No Title Lines Possible with Filters and/or Frozen Top Row 

Columns are AutoFit, but filter buttons partially overlay headings 



ods results; /* MUST BE TURNED ON so you can get 

and respond to prompts */ 

ods graphics / reset=all; ods graphics off;  

goptions reset=all; 

ods tagsets.TableEditor style=minimal 

file="%sysfunc(getoption(work))\temp.html" 

  options( sheet_name="Frozen Headers and Filters" 

excel_orientation="landscape" 

excel_frozen_headers="yes“ 

excel_autofilter="yes" 

auto_excel="yes" 

excel_save_file=“C:\\SomeFolder\\SomeFile.xls" ); 

 /* Note double back slash \\ in path above. */ 

title1 'This title text can not be displayed.';  

proc print data=sashelp.shoes noobs; run;  

ods tagsets.TableEditor close; 



     

   Nice-To-Have Features in Excel 
- Multi-Sheet WorkBooks 

- Multiple Tables in WorkSheet 

- Multiple Graphs in WorkSheet 

- Graphs and Tables in WorkSheet 

- Possible with MSOffice2K_x tagset, 

  which also supports panels of objects 

- Most flexibility with DDE 
 

Here, Panel = one row of side-by-side cells,  

to improve ability to see all tables/graphs 

at the same time 
 



 Multi-Sheet Workbooks and 

 Composite WorkSheets 
x 



 Multi-Sheet Workbooks and 

 Composite WorkSheets 
  

• ODS HTML 

• single-worksheet workbooks only  

• no panels, multiple outputs are stacked 



 Multi-Sheet Workbooks and 

 Composite WorkSheets 
  

• ODS HTML 

• single-worksheet workbooks only  

• no panels, multiple outputs are stacked 

• ODS tagsets.ExcelXP 

• multi-sheet workbooks, no panels, no graphs 



 Multi-Sheet Workbooks and 

 Composite WorkSheets 
  

• ODS HTML 

• single-worksheet workbooks only  

• no panels, multiple outputs are stacked 

• ODS tagsets.ExcelXP 

• multi-sheet workbooks, no panels, no graphs 

• ODS tagsets.MSOffice2K_x 

• multi-sheet workbooks, panels, graphs 



 Multi-Sheet Workbooks and 

 Composite WorkSheets 
  

• ODS HTML 

• single-worksheet workbooks only  

• no panels, multiple outputs are stacked 

• ODS tagsets.ExcelXP 

• multi-sheet workbooks, no panels, no graphs 

• ODS tagsets.MSOffice2K_x 

• multi-sheet workbooks, panels, graphs 

• ODS tagsets.TableEditor 

• multi-sheet workbooks, panels, graphs 

• most interesting functions not viable for batch 

  (require reply to prompts in Display Manager) 



Best Non-DDE Choice 

MSOffice2K_x 

if you want these features: 
• multi-sheet workbooks 

• panels 

• graphs 



ODS tagsets.MSOffice2K_x 
  

Horizontal Panels 

• WorkSheet with Multiple Graphs 

• WorkSheet with Multiple Tables 

• WorkSheet with Multiple Tables and One Graph 

  (In a panel of tables and graphs, 

  one graph is the maximum, 

  and it must be last in the panel) 

If you omit the panel option, 

objects are presented one above the other, 

with no limit 

on the number of graphs or tables, 

and they can be presented in any order. 



First, a Simple Case: 

Table at Left of Graph 

(and Sheet Name Customization) 

 

 

 

 





* Table at Left of Graph *; 

* (and Excel Sheet Name Customization) *; 
 

ods tagsets.MSOffice2K_x style=Minimal 

  path="C:\SomeFolder"  

  file="TableAtLeftOfGraph.xls"  

  options(panelcols="2" /* panel_space="1" */ 

               sheet_name="Table At Left Of Graph"  

               doc="help"); 

< proc print step here > 

< proc gchart step here > 

ods _all_ close; 
 

* doc="help" to list all of the tagset options in SAS log *; 

 

 



           MSOffice2K_x panels 

panel = one row 

of tables 

or graphs 

or tables and one graph 

 

 

 

 

 



           MSOffice2K_x panels 

. graphs only – any number OK 

. tables only – any number OK 

. any number of tables 

  at left of one graph OK 

 

 

 

 

 



 

 extra graphs (>1) at right of table(s) 

 will overlay each other 

 

 

 

 

 

 

 



Problem: graphs overlay if > 1 at right of table(s) 



 

   table at right of graph goes 

   below graph, not next to it 

 

 

 

 

 

 

 



Problem: table desired at right goes below graph 



Multi-Sheet Workbook using 

MSOffice2K_x TagSet 
  

• MSOffice2K_x WorkBook is a collection of files 

• See discussion later about keys to Portability* 

• Use a SAS macro 

  if you wish to simulate the BY processing 

  option available when using ExcelXP tagset 

*Portability = 

can move the collection to another folder, 

or email it, 

and everything will still work. 



MSOffice2K_x WorkBook is a collection of files 



      Multi-Sheet Workbook using 

           MSOffice2K_x TagSet 

Several Sheets: 
  

• Three Tables Side-by-Side 

• Three Plots Side-by-Side (three ways): 

• SGPLOT with SCATTER statement 

• GPLOT with PLOT statement 

• SGSCATTER with PLOT statement (defect) 

• Two Tables and One Plot Side-by-Side 

• Two Univariate Reports on Separate Sheets 

• Extra Sheet (demo only) – Warnings 



      Multi-Sheet Workbook using 

           MSOffice2K_x TagSet 

More Sheets: 
  

• Table Over Plot 

• Two Tables Over Two Plots 

• Plot Over Table 

• Two Plots Over Two Tables 



Let’s Open the Result 



 

 

 

There is no corruption 
 

Get this with every ODS tagset 

if you create Excel output. 

Exception: Create ExcelXP with 

filetype xml. 



 

 

 

 

 

 

 

This is a False Alarm 

No Problem 
 



This is a False Alarm 

No Problem 
 

But if you omit all graphs, 

or create one per sheet 

or multiple, but stacked within the sheet, 

(i.e., avoid a side-by-side panel), 

the pop-up does not occur. 

 

 

 
 

 



For Demo Only 
 

My workbook includes a warning worksheet. 
 

You MIGHT want to retain the second 

warning message in a warning worksheet if 

any other worksheets in the workbook are 

panels of graphs. 

 

 

 

 

 
 



WorkSheet of Warnings 



PROC PRINT steps deliver panel of 3 tables - zoom=“90” 



ods tagsets.msoffice2k_x path="C:\SomeFolder“ 

  file=“ThreeTables.html" style=minimal 

  options( orientation="landscape" zoom="90" 

  panelcols="3" panel_space="1" ); 
 

proc print data=sashelp.class; 

title color=red " title text here "; 

var height age; 

run; 
 

 /* like above for two more tables */ 
 

ods tagsets.msoffice2k_x close; 

 

 

 

 

 



PROC SGPLOT SCATTER steps deliver panel of 3 images - zoom=“85” 



goptions reset=all; 

ods graphics on / reset=all height=4in width=4in 

border=on antialiasmax=2500; 
 

ods tagsets.msoffice2k_x path="C:\SomeFolder"  

 (url=none) file="SGplots.html" style=minimal 

  options( zoom="85" orientation="landscape" 

  panelcols="3" panel_space="1" graph_width=“300" ); 
 

ods graphics on / 

       imagename='HgtVsAge_SGPLOT_SCATTER';  

proc sgplot data=sashelp.class; 

title color=red " title text here "; 

scatter y=height x=age; run; 
  

 /* like above for two more plots */ 
 

ods tagsets.msoffice2k_x close; 

 



NOTE: In an earlier version of this code, 

I had used a smaller height & width = 3in 

and graph_width = 225 and default zoom = 100. 

With those parameters, three plots did fit in the Excel 

worksheet window, and there was no need to scroll 

to the right to see the last plot. 

The monitor resolution was set to 1024 X 768. 
 

ods graphics on / reset=all height=4in width=4in 

border=on antialiasmax=2500; 
 

ods tagsets.msoffice2k_x path="C:\SomeFolder"  

  (url=none) file="SGplots.html" style=minimal 

  options( zoom="85" orientation="landscape" 

  panelcols="3" panel_space="1" graph_width=“300" ); 

 

 



PROC GPLOT PLOT steps deliver panel of 3 images - zoom=“85” 



PROC SGSCATTER PLOT –  images overlaid when XLS file is viewed in Excel 



Two Tables at Left of Plot (zoom=“95”), plot must be last, maximum one plot 



goptions reset=all; 

ods graphics on / reset=all height=3in width=3in 

border=on antialiasmax=2500; 

ods tagsets.msoffice2k_x path="C:\SomeFolder“ 

 (url=none) file=“TwoTablesOnePlot.html" style=minimal 

  options( orientation="landscape" zoom= "95"  

  panelcols="3" panel_space="1" graph_width="225" ); 

< proc print step > 

< proc print step > 

< proc sgplot step > 

ods tagsets.msoffice2k_x close; 

 

 

 

 

 



proc print data=sashelp.class; 

title “Name & Height"; var name height; run; 

 

proc print data=sashelp.class; 

title “Name & Weight"; var name weight; run; 

 

ods graphics on / imagename=‘WgtVsHgt_2';  

proc sgplot data=sashelp.class; 

title color=red “Weight vs Height"; 

scatter y=height x=age; run; 

 

 

 



Table Over Plot (zoom=“80”) 



Two Tables Over Two Plots (zoom=“50”) 



Two Tables Over Two Plots (zoom=“35”) 



Plot Over Table (zoom=“80”) 



Two Plots Over Two Tables (zoom=“50”) 



Two Plots Over Two Tables (zoom=“35”) 



PROC UNIVARIATE step delivers serial chain of tables for height statistics 



ods noproctitle; 
 

ods tagsets.msoffice2k_x path="C:\SomeFolder" 

  file="HgtStats.html" style=minimal; 
 

proc univariate data=sashelp.class; 

title; 

var height; 

run; 
 

ods tagsets.msoffice2k_x close; 
 

 /* same for var weight */ 

 

 

 

 

 

 



PROC UNIVARIATE step delivers serial chain of tables for weight statistics 



As Of 18 October 2013, 

using SAS Version 9.4 (TS1M0) 

for the two preceding PROC UNIVARIATE examples 

the note “The mode displayed is the smallest of 2 

modes with a count of 2.” 

is not correctly displayed 

with the MSOffice2K_x tagset.  

For how to circumvent this problem, 

at support.sas.com see SAS Note 51405. 



ods tagsets.msoffice2k_x  

  path="C:\SomeFolder" file="SomeFile.xls" 

  options(worksheet_source="  

  Warn#.\WarningMessageFromLeRB.html, 

  3Tables#.\ThreeTables.html, 

  3SGplots#.\SGplots.html, 

  3Gplots#.\Gplots.html, 

  3SGscatters#.\SGscatters.html, 

  Tables+Plot#.\TwoTablesOnePlot.html, 

  HgtStats#.\HgtStats.html, 

  WgtStats#.\WgtStats.html 

  ") ; 

data _null_; file print; put _all_; run;                                                                                     

ods tagsets.msoffice2k_x close; 

 

 



About dot . in Tables#.\ThreeTables.html 

- The dot is signal that html file is in the same 

folder as that of the output XLS file. 
 

- The (url=none) in graph ODS steps is a signal 

that any image file(s) will be in the same folder as 

the HTML file. 
 

- These codings assure that package of xls 

file, html files, and image files is relocatable: 

-- Move anywhere 

-- Send via email 

-- XLS file will open & all parts will be found 

 

 



About dot . in Tables#.\ThreeTables.html 

- The dot is really the usual  round dot 

- My choice of font renders it 

  as a square in this slide. 

 

 

 

 

 

 

 

 

 

 



ODS tagsets.MSOffice2K_x 
  

MultiSheet WorkBook with WorkSheets 

containing tables, graphs, tables and graphs. 

Objects can be presented in a horizontal 

panel, or stacked one above the other. 

Only limit: 

in a horizontal panel of tables and graphs, 

maximum of one graph 

and it must be presented last. 

Only combination not tested: 

stack with alternating tables and graphs. 

My conjecture: Unlikely to be a problem. 

If you find otherwise, please notify me. 



Multi-Sheet Workbook using 

ExcelXP TagSet 
  

• No support for graphs 

• No panels 

• Can exploit SAS BY processing 

• WorkBook is a single file 

• Easy multi-sheet WorkBook creation 



easy multi-sheet workbook creation 

using ExcelXP tagset 
 < usual setup code goes here > 

ods tagsets.ExcelXP path="C:\Folder" 

  file="FileName.xml" style=minimal 

  options(embedded_titles="yes" wraptext="no"); 

ods tagsets.ExcelXP options(sheet_name="Girls"); 

proc print data=sashelp.class(where=(Sex EQ 'F')) noobs; 

title1 justify=left height=16pt color=red 'text 1'; run; 

ods tagsets.ExcelXP options(sheet_name="Boys"); 

proc print data=sashelp.class(where=(Sex EQ 'M')) noobs; 

title1 justify=left height=16pt color=blue 'text 2'; run; 

ods tagsets.ExcelXP close; 







/* Can Exploit SAS BY Processing */ 

< usual setup code goes here > 

ods tagsets.excelxp style=Minimal path="..." file="..." 

  options(embedded_titles="yes" wraptext="no"); 

ods tagsets.excelxp options(sheet_name="Height"); 

title1 justify=left height=16pt 'Height of Students' 

proc print data=sashelp.class noobs; 

id name; var height; run; 

ods tagsets.excelxp options(sheet_name="none"); 

ods tagsets.excelxp options(sheet_interval='bygroup' 

  sheet_label='Sex' suppress_bylines='yes'); 

proc sort data=sashelp.class out=Sorted; by sex; run; 

title1 justify=left height=16pt 'Age of Students'; 

proc print data=Sorted noobs; 

by sex; id name; var age; run; 

ods tagsets.excelxp close; 









For More on ODS TagSets 
  

In Proceedings and Tools: 

• Other ODS SAS-to-Excel Resources— 

  Tips, References, Sample Program SetUp 



DDE 



What is DDE? 

Dynamic Data Exchange: 

A concurrent Excel session 

is started by the SAS session. 

The Excel session acts as a server 

for the SAS session. 

The SAS session acts as a client of Excel. 

SAS & Excel must run 

on the same Windows computer .  



What Can You Do with DDE? 
  

• Anything you can do with keyboard & mouse 

• Commands simulate mouse & keyboard actions 

• Now let’s look at how use of DDE is coded 



 

options noxwait noxsync; 

 
x ' "C:\Program Files (x86)\Microsoft Office\Office14\EXCEL.exe" '; 

  /* this path depends on your version of Windows and of Excel */ 
 

data _null_; 

z=sleep(3); /* wait 3 seconds for Excel to start */ 

run; 
 

 /* Above code launches Excel and opens a new workbook, 

    with one empty worksheet: Sheet1. */ 
 

 /* After opening the empty worksheet, 

    the next step could be any one of these: 

    - Format the empty cells 

    - Open an existing workbook instead 

    - Use DDE to load the empty worksheet */ 

 

 

 





 

filename target dde 

  'excel|Sheet1!R1C1:R999C6' notab; 
 

 /* to load data in columns 1 to 6 

    of rows 1 to 999. 

    The ending row number must equal or exceed 

    the number of SAS observations to be sent to Excel. */ 
 

data _null_; 

set sashelp.shoes; 

file target; 

 < PUT statements here to load headings row and data rows > 

run; 

 

 

 

 

 

 



data _null_; 

set sashelp.shoes; 

file target; 
 

if _N_ EQ 1 then do; 

  put 'Region'  '09'X   'Subsidiary'  '09'X   'Product'  '09'X 

        'Stores'  '09'X   'Sales'  '09'X   'Inventory'  '09'X   'Returns'; 

end; 
 

put Region '09'X Subsidiary '09'X   Product '09'X 

      Stores '09'X   Sales '09'X   Inventory '09'X   Returns; 
 

run; 
 

 /* '09'X is a delimiter that tells Excel 

    to step to the next cell in the current worksheet row 

    before inserting next variable from current SAS observation */ 

 

 





filename DDEcmds dde "excel|system"; 

 /* to send commands to Excel from SAS */ 
 

 /* In the steps that follow, 

only one or few DDE commands 

are issued in each DATA step.  
 

The steps could be consolidated into one. 
 

Here, the steps are kept discrete, 

for the sake of simplicity and brevity, 

and to be able to display the evolution 

of formatting of the worksheet 

by each successive command. */ 
 

 

 



filename DDEcmds dde "excel|system"; 

 /* to send commands to Excel from SAS */ 

 

data _null_; 

file DDEcmds; 

 

put '[select("C1:C7")]'; 

put '[font.properties("Arial",,16,,,,,,,1)]'; 

 /* Columns 1-7 Arial 16pt Black */ 

 

put '[select("R1")]'; 

put '[font.properties(,"Bold")]'; 

 /* Row 1 Bold */ 

 

run; 

 

 





data _null_; 

 

file DDEcmds; 

 

put '[select("C1:C4")]'; 

 /* only contiguous groups of columns 

    can be used with filters */ 

 

put '[Filter()]'; 

 

run; 

 

 

 

 

 

 





data _null_; 

 

file DDEcmds; 

 

put '[freeze.panes(true,3,1)]'; 

 

 /* Between ,,  (if not the value 0), 

    put number of rightmost frozen column.                                    

    After ,, (if not the value 0), 

    put number of bottommost frozen row. */ 

 

run; 

 

 

 

 

 





data _null_; 

 

file DDEcmds; 

 

put '[select("C1:C6")]'; 

 

put '[column.width(,,,3)]';  /* 3 = AutoFit */ 

 

run; 

 

 

 

 

 

 

 

 





data _null_; 

 

file DDEcmds; 

 

put '[zoom(70)]'; 

 

put '[message(true, 

                        "This sheet is reduced to 70% of normal size")]'; 

 

run; 

 

 

 

 

 

 

 





Let’s magnify the message at the lower left 





data _null_; 

 

file DDEcmds; 

 

put '[select("R1")]'; /* insert four blank lines above row 1 */ 

 

put '[insert()]'; 

put '[insert()]'; 

put '[insert()]'; 

put '[insert()]'; 

 

run; 

 

 

 

 

 





data _null_; 

 

file DDEcmds; 

 

put '[select("R1")]';  

put '[font.properties(,"Bold",,,,,,,,5)]';   /* Row 1 Bold Blue */ 

 

put '[select("R2")]'; 

put '[font.properties(,"Bold",,,,,,,,3)]'; /* Row 2 Bold Red */ 

  

put '[select("R3")]'; 

put '[font.properties(,,,,,,,,,5)]'; * Row 3 Normal Blue */ 

 

put '[select("R4")]'; 

put '[font.properties(,,,,,,,,,1)]'; /* Row 4 Normal Black */ 

 

run; 

 



Four rows of cells formatted, but effect not yet apparent. 



filename titles dde 'excel|Sheet1!R1:R4' notab; 
 

data _null_; 

 

file titles;   /* load title text into rows 1 to 4 */ 

 

put 'Listing of SASHELP.SHOES . . .'; 

put 'To visit the . . .'; 

put 'http://www.wiilsu.org'; 

put 'The only disadvantage . . .'; 

 

run; 

 

 

 

 

 

 





data _null_; 

 

file DDEcmds;  

 

put '[select("R1")]'; 

  /* Move focus from row 4 after the preceding title fill step */ 

 

run; 

 

 

 

 

 

 

 

 

 





 

 /* Finished Workbook Save As */ 

 

data _null_; 

 

file DDEcmds; 

 

put '[error(false)]';  /* suppress any prompt questions */ 

 

put 

'[save.as("D:\Folder\DDE_Customized_WorkSheet_Of_SASHELP

dotSHOES.xls")]'; 

 

x=sleep(3); 

 

run; 

 

 

 



Note the “Saved As” filename below: 



 /* Exiting Excel from SAS */ 

 

data _null_; 

 

file DDEcmds; 

 

put '[error(false)]';  /* suppress any prompt questions */ 

 

put '[quit()]';  /* the parenthesis is empty */ 

 

run; 

 

 

 

 

 

 

 



Can You Use DDE? 



Running DDE on your own PC 
You can go to the Excel window at each step of 

the program to inspect the result of commands 

submitted so far. 

If not satisfied with the current state, you can 

manually use Excel UnDo, modify your SAS 

code, and resubmit the code. 

If SAS and/or Excel are/is hung, you can 

manually intervene (or use Windows Task 

Manager, if needed). 

If running SAS on your PC via Enterprise Guide, 

EG might already be enabled to submit code that 

launches Windows commands. 



Possible Special Need #1 for DDE 
If running SAS on your PC via Enterprise Guide, 

EG might already be enabled to submit code that 

launches Windows commands. 
 

The SAS X command is used to launch 

the Windows command that starts Excel. 
 

If you get this message 
ERROR: Shell escape is not valid in this SAS session. 

see 

http://blogs.sas.com/content/sasdummy/2009/11/

19/using-the-x-and-systask-commands-from-sas-

enterprise-guide/  



Possible Special Need #2 for DDE 
If running SAS on a server,  SAS there must be 

enabled to submit code that launches Windows 

commands. 
 

The SAS X command is used to launch 

the Windows command that starts Excel. 
 

If you get this message 
ERROR: Shell escape is not valid in this SAS session. 

ask your server administrator to turn on the 

XCMD system option  



Possible Inconvenience of DDE 
 

If SAS runs on a remote server, not on your PC, 

and a malfunction occurs: 

user needs ability to kill a hung SAS process 

user needs ability to kill a hung Excel process 

user needs ability to identify the Process IDs 

OR user must get SAS administrator to intervene. 

For above needs, I have provided user tools for 

SAS (which can be adapted for Excel). See: 
http://support.sas.com/resources/papers/proceedings10/279-2010.pdf 



For More on DDE 
  

In Proceedings and Tools: 

• the code, log, and output for these DDE slides 

• my expansive DDE paper and DDE toolkit 



NOT In Proceedings & Tools 
I learned SAS coding for DDE 

from two papers by Koen Vyverman: 

• http://www2.sas.com/proceedings/sugi26/p011-26.pdf 

• http://www2.sas.com/proceedings/sugi27/p190-27.pdf 

See also the SAS-provided documentation. 

In SAS, click on SAS Help and Documentation 

> Using SAS Software in Your Operating Environment  

> SAS 9.X Companion for Windows 

(where 9.X is YOUR version of SAS) 

> Using SAS with Other Windows Applications 

> Using Dynamic Data Exchange Under Windows 



 Feature  TableEditor*  ExcelXP  MSOffice2K_x     HTML       DDE** 

 Pivot Table  Yes  No  No  No 

 requires an 

Excel macro 

 AutoFit 

 Columns  By Default*   No  By Default* By Default  Yes 

 Filters 

 Yes, but All 

 Columns only  Yes 

 Yes, but All 

 Columns only  No  Yes 

 Freeze Panes  Row 1 only  Yes  Yes  No  Yes 

 Side-By-Side 

 Elements in 

 WorkSheet  Yes  No  Yes  No  Yes 

 Multi-Sheet 

 WorkBook  Yes  Yes, Easy  Yes  No  Yes 

 Graphs  Yes  No  Yes  Yes  Yes 

 Titles 

 Not with 

 other features   Yes  Yes  Yes  Yes 

Comparison of Tool Alternatives 

*Headings of columns autofit by default can be overlaid by filter buttons 

*Javascript must be enabled **SAS must be running on Windows 



Conclusion 

 Dynamic Data Exchange (DDE) is very powerful, 

but if you use SAS on a server, esp. via SAS 

Enterprise Guide, you can have awkward 

situations that you need to be able to deal with. 
 

Kevin Smith at SAS Institute is working on an 

ODS Excel destination intended to deliver more 

function than any of the existing ODS solutions. 
 

An excellent new SAS Institute course: 

Exporting SAS Data Sets and Creating ODS 

Files for Microsoft Excel 



About SAS Institute Training 

Exporting SAS Data Sets 

and Creating ODS Files for Microsoft Excel 
 

Exported data is unformatted 
 

DDE tool not included in the course 
 

TableEditor is self-study only 

